



Coco Nutz!

Volume 1, Issue 1, Jan 5, 2005

Index

Contact with the other side

Interview of the month

List of keyboard commands

How to set up your CoCo

Splash and Run

Basic programs

Make a donation to the Newsletter! <http://tinyurl.com/6gphh>

Contact with the other side.

What sparked my interest in the Trs-80 color computer was my Dad. After he died I received some disks; well actually about 500 5¼ inch floppy disks. I didn't have a computer to run them on, as that sort of disk drive is older and most computers do not come with that size disk drive anymore. So my first task in finding out what was on those disks, and how they could tell me more about my father, was to locate the hardware.

I first went to some websites and sent out emails to everyone I could find. The first person that answered was Roger Taylor. He then introduced me to his website where a bunch of CoCo users go to. Then he told me how to get on a coco mailing list with about 200 members on it. So I got on that list and sent out an email asking anyone who might have the equipment I need to sell it to me. The response I got brought me to tears!

First to contact me was Mannequin, he offered up a Tandy trs80 computer for free. Then up was Jim and he offered up various disk drives, again for free. Then up was a man that is like a father to me, Mr. Bob whom offered me other hardware such as monitor again for free. (Bob also offered to pay shipping on all of these items before I could even say no!) The people on this coco mailing list have touched my heart deeply. Without hesitation and just as fast as I could ask these men took me into their lives like a daughter and were willing to help me with my quest to find out who my father was.

So to Mannequin and his family, Uncle Jim Cox, and Daddy Bob, and of course Roger Taylor, for leading me to these wonderful people, Thank you again so much.

Interview of the month

Bob Devries



This is a little part of the newsletter where I interview someone from the CoCo community. This time is going to be Mr. Bob Devries. Bob is a student in a small country town of Dalby, Queensland, Australia. He is currently studying for his certificate in Christian ministry. "I hope to specialise in counseling." He says. When I asked him what he does in his free time he said, "Hehehe, free time? Well, I spend a lot of time chatting to a certain lady on the cocochat....and a bunch of really great guys there too. As well as that I also do Christian counseling on-line."

He started working with the coco through his friends at work. They had actually built a coco clone from the schematics published by Motorola in their handbook. Bob has written a number of utility programmes and even contributed in a small way on the current nitros-9 project.

He was also part of something much bigger. He took over the os-9 newsletter in Sydney when the publisher asked him to. Graham Nichols started it off, and then in early 1988 he decided he could no longer keep it going. He asked Bob, Gordon Bentzen, and Don Berrie if they would take it over. They started off with his original mailing list, and it sort of grew from there. They published 11 issues per year.

They added other guys to their "staff", and also set up an Australian OS9 software library. Bob, Gordon, and Don continued until August 1994. Interest in the newsletter then sort of waned, and got to where it was no longer economical to do. They did only OS9 stuff, but printed articles of interest about hardware, software reviews, and of course, programme listings in Basic09, C, and Assembler. They didn't only cover OS9 for the Coco, either. The trio spread out to the 68000 version when the MM/1 came out, and also mentioned the Atari and other OS9 versions. During that time, Bob built his own 68000 based computer from a kit, The SECAD AS-68K, and it came with OS9/68000 Professional. Bob was then an electronics technician. He built, repaired, and programmed computers for a hobby....and still does a bit. "I was actually able to get my first Coco3, because I upgraded a number

of Coco2 units for the local Tandy store. I upgraded 10 16k coco2 computers to 64K, and in exchange got a 128K coco3. That was in '86 when they came out. Along with the coco3 of course, I got into OS9 Level 2.... and never looked back. Yeah. I love them so much because they are virtually indestructible, and easy to programme. " he said.

Bob Devries can be found online most of the time or in the coco3.com chat room. So stop in and say HI to him and tell him you read all about him in this issue of the newsletter!

Keyboard Characters and Symbols

ENTER tells the computer you've ended a line

BREAK stops running program

CLEAR clears the screen and returns the cursor to the upper left corner

SHIFT @ pauses your running program. Press any key to continue

SHIFT "LEFT ARROW" erases an entire line if ENTER has not been pressed

"LEFT ARROW" backspaces the cursor and erases one character at a time

? means the same as the command PRINT

: separates statements on a line

; tells the computer to continue PRINTing on the same line

+ addition, and also combines strings

- subtraction

x or * multiplication

/ division

() tells the computer to do what's inside first

Basic Commands

CLS clears the screen with green

CLS () clears the screen with the color indicated by the number inside the parentheses e.g. CLS 5

CLOAD loads a program from tape

CSAVE saves a program to tape

END tells the computer to end your program

FOR/NEXT creates a loop in a program that the computer will repeat the number of times

GOTO sends the computer to the line number indicated

LIST lists a program on the screen

LLIST lists a program on the printer

PRINT tells the computer to print whatever follows

PRINT @ prints at a specific screen location

PRINT TAB indents the number of spaces indicated within parentheses

PRINT # - 2 prints on the printer

REM a remark statement that the computer ignores

RESET resets a point to black

SET sets a point on the screen at the position and color specified

SOUND sounds a specified tone for the duration of time specified

STEP causes the for/next loop to count by the number that follows the word STEP

How to set up a coco and get it running

(from a newbie's point of view)

Setting up a coco for me sounded pretty easy. I figured you just sit it down and plug it in; well its a little bit more complicated than that.

So here is my account of what I had to do to get it up and running. I had to find a stable place to put it. I knew I had to give it its own desk so that the other components would sit near it comfortably. I also had to make sure my desk was deep enough to set the keyboard in front of the monitor with a little distance between the two. What sits where? I had to think of how to position things as well. I put my monitor in the middle of the desk with the keyboard in front of it. Then I put my MPI and disk drives to the right side of the keyboard. I later moved the disk drives to the left of the computer since the ribbon was long enough to do so.

Ok now I had to plug it all up. Everything was pretty self explanatory like the power cords. I had to figure out that the monitor plugged into the underside of the computer and that the mouse plugs into the right joystick plug in the back of the coco. The MPI or floppy controller only plugged into one spot so that was easy to figure out as well. It plugged into the right side of the computer. Then I went to plug it all into an outlet and soon realized I had only two spots. So I invested in a power strip to plug all the power cords into. There were 4 power cords, one for the monitor, the drives, the computer, and the MPI.

Now before I turned mine on I asked some coco users for any precautions on how to do so.

These are some of the comments I received:

MAKE SURE the hardware is secure and not wobbling.

MAKE SURE the MPI or the disk controller is securely attached to the computer.

MAKE SURE there are no disks in drive and that the drive doors are open.

TAKE A MINUTE to look at where all the power switches and the reset button are on all the hardware.

MAKE SURE no cords are in the way of your feet or chair etc.

MAKE SURE there is nothing blocking the doors on the drives.

Ok now comes the part when I turned it on. There is a very extremely important power on sequence. First you turn on the Monitor, Then the MPI, then the Drives, and always turn on the computer last! You do the opposite to turn off the system. First turn off the computer, then the drives then the MPI and then the monitor. You can hit the reset button if you want to stop a program and restart your computer. Just for fun try this. With the computer on hold down the CTRL and ALT on the keyboard, and RESET button on the back of the computer, then let go of the reset button while still holding down the CTRL and ALT keys, then let go of them and you will see a picture of three men. I was told that these guys

designed the software for the coco 3.

There are a few no no's I learned about my coco that could cause permanent damage to it:

DO NOT ever move the MPI or the computer around while it is turned on.

DO NOT scoot it back and forth on the desk or anything it will apparently fry your motherboard.

NEVER unplug or plug anything into the computer while it is turned on.

DO NOT turn the computer off or on with a disk in the drives.

MAKE SURE the drives are empty before turning them on, or at least the drive door is open.

Now you got the thing on and don't know what to do with it. Here are some of my first basic commands I learned. I learned how to change the screen color using CLS 1-8. Just type CLS and a number and the screen changes colors. Example is CLS 2. To get it back to command prompt color type: CLS 1. I learned how to make sounds using the "sound" command. For example type "SOUND 1, 100" without the quotes. I also learned that the coco is a calculator. Try this math problem. Type PRINT 1589 * 23 and press enter. You can also add, divide and subtract using the same format. If you have the book "Getting Started with Extended Color Basic". it will tell you all you need to know to get started.

Well I hope I have given you all you need to get your Color Computer 3 set up and ready for some programming. So have fun with it.


```
FCC " "  
FCC " "  
FCC " "  
FCC " "  
FCC " "  
FCC " "  
FCC " "  
FCC " "
```

* Here's the actual program code that gets executed,
* which is nothing more than an endless loop.

```
ORG 3584  
START BRA START    loop forever
```

* This section causes the Disk BASIC 'LOADM' command
* to start loading the subsequent code right into the
* system memory at 356, which just happens to be some
* vectors that are called by BASIC when the LOADM
* command is finished.

```
ORG 356  
LBRA START    replace 3 vectors with our  
LBRA START    our own code that jumps  
LBRA START    right into our program!  
  
END START
```

Basic programs

This is a name and address programme that uses random access filing method that is part of DISK BASIC. Like a little address book, you can search for an entry, add new ones, delete records etc. It actually has a HELP section in it, too. It was written for Australian conditions, so be aware that the postcode field is only 4 characters long. The programme will run on a coco3, but it was designed to run on the 32 column screen (I.E. WIDTH32) questions can be directed to my email address: bdevries@gil.com.au The phone number listed in the program is obsolete.

```
10 'SIMPLE ADDRESS DATABASE USING DIRECT ACCESS FILING
20 GOTO 50
30 SAVE"DATABASE":DIR:PRINTFREE(0)
40 END
50 CLS:CLEAR500:RECORD=0
60 PRINT@3,"SURNAME:-----:";
70 PRINT@33,"FIRSTNAME:-----:";
80 PRINT@68,"STREET:-----:";
90 PRINT@102,"CITY:-----:";
100 PRINT@129,"POSTCODE:----:";
110 PRINT@143,"PHONE:-----:";
120 PRINT@449,"RECORD####";:PRINT@472,"LOF
###";:PRINT@480,"SEARCH eDIT gOTO pRINT hELP qUI";:POKE&H5FF,&H54
130 GOSUB1010
140 PRINT@456,USING"###";RECORD;
150 PRINT@476,USING"###";LF;
160 KEY$=INKEY$:IF KEY$=""THEN160
170 KEY=INSTR("PEGHSQ",KEY$)
180 ONKEY+1 GOSUB 210,240,410,540,650,760,220
190 GOSUB 590
200 IF KEY$="H" THEN GOTO 50 ELSE GOTO 160
210 RETURN
220 CLS:END
230 'PRINT ONE RECORD OR ALL RECORDS
240 PRINT@420,"RECORD # OR
aLL";:INPUTR$:PRINT@420,STRING$(19,32);
250 IF R$="" THEN RETURN ELSEIF LEFT$(R$,1)="A" THEN 360
260 R1=VAL(R$):IF R1=0 THEN GOSUB600:GOTO240 ELSE IF R1>LF THEN
GOSUB600:GOTO 240
270 RECORD=R1:GOSUB1190
280 PRINT#-2," SURNAME: ";N1$
290 PRINT#-2,"FIRSTNAME: ";N2$
300 PRINT#-2," STREET: ";S$
310 PRINT#-2," CITY: ";C$
320 PRINT#-2," POSTCODE: ";P1$
330 PRINT#-2," PHONE: ";P2$
340 PRINT#-2,CHR$(13);CHR$(13)
```

```

350 RETURN
360 FOR X=1 TO LF
370 R1=X:GOSUB270
380 NEXT X
390 RETURN
400 'EDIT RECORD
410 GOSUB420:GOTO490
420 GOSUB520:LINEINPUT"SURNAME:";N$:IFN$<>" THEN N1$ =
LEFT$(N$+STRING$(20,32),20)
430 GOSUB520:LINEINPUT"FIRSTNAME:";N$:IFN$<>" THEN N2$ =
LEFT$(N$+STRING$(20,32),20)
440 GOSUB520:LINEINPUT"STREET:";N$:IFN$<>" THEN S$ =
LEFT$(N$+STRING$(20,32),20)
450 GOSUB520:LINEINPUT"CITY:";N$:IFN$<>" THEN C$ =
LEFT$(N$+STRING$(20,32),20)
460 GOSUB520:LINEINPUT"POSTCODE:";N$:IF N$<>" THEN P1$ =
LEFT$(N$+STRING$(4,32),4)
470 GOSUB520:LINEINPUT"PHONE:";N$:IF N$<>" THEN P2$ =
RIGHT$(STRING$(10,32)+N$,10)
480 RETURN
490 GOSUB1060:GOSUB890
500 GOSUB520:PRINT@270,"CORRECT ?";
510 K$=INKEY$:IFK$=""THEN510 ELSE IF K$="Y" OR K$="y" THEN
PRINT@270," ";:GOTO 130 ELSE 410
520 PRINT@256,STRING$(31,32):PRINT@257,"";:RETURN
530 'GOTO RECORD NUMBER
540 PRINT@420,"";:INPUT"RECORD #";R1
550 IF R1=LF+1 THEN GOSUB 970:GOTO580 ELSE IFR1>LF OR R1=0 THEN
600
560 RECORD=R1
570 GOSUB1190:'GET RECORD
580 PRINT@456,USING"###";RECORD;:GOSUB 890:'PRINT RECORD TO
SCREEN
590 PRINT@420,STRING$(14,32):RETURN
600 PRINT@420,"INVALID RECORD";
610 FOR TD=1 TO 300:NEXTTD
620 PRINT@420,STRING$(14,32)
630 RETURN
640 'HELP SCREENS
650 CLS:PRINT@8,"MINI DATA BASE":PRINT@40,"BY BOB DEVRIES"
660 PRINT:PRINT" THIS IS A VERY SIMPLE DATABASE AND ITS FEATURES
ARE SIMPLE TO UNDERSTAND.":PRINT" S. SEARCHES THE DATABASE FOR
AN ENTRY. YOU MAY USE ONE OR ALL FIELDS, AND LEAVE THE UNUSED
ONES BLANK."
670 PRINT" E. EDITS THE CURRENT RECORD THAT IS THE ONE
DISPLAYED, BUT USE THE GOTO COMMAND FIRST. IF YOU YOU DON'T
WANT TO CHANGE A FIELD JUST PRESS ENTER.":PRINT" PRESS ANY KEY TO
CONTINUE";
680 I$=INKEY$:IFI$=""THEN680

```

```

690 CLS:PRINT" G. GOTO A RECORD, RECORD NUMBER MUST LESS THAN OR
EQUAL TO LOF OR TO INSERT A NEW RECORD, GOTO LOF + 1. E.G. IF
LOF=3 THEN GOTO 4 WILL CREATE A NEW BLANK RECORD FOR YOU TO
EDIT.":PRINT" P. PRINT RECORDS. YOU WILL BE";
700 PRINT" PROMPTED FOR RECORD NUMBER OR ALL. IF 'A' IS
CHOSEN ALL THE FIELDS WILL BE PRINTED ON THE PRINTER"
710 PRINT:PRINT" DIRECT ANY QUESTIONS TO: BOB DEVRIES PH
3727816"
720 PRINT@481,"PRESS ANY KEY TO CONTINUE";
730 I$=INKEY$:IF I$="" THEN 730
740 RETURN
750 'SEARCH FOR ENTRY
760 N1$="":N2$="":S$="":C$="":P1$="":P2$="":X=0:GOSUB420
770 N3$=N1$:N4$=N2$:S1$=S$:C1$=C$:P3$=P1$:P4$=P2$
780 X=X+1:IF X>LF THEN PRINT@420,"NOT FOUND";:GOSUB520:FORTD=1 TO
50:NEXT TD:RETURN
790 RECORD=X:GOSUB1190
800 IF N3$="" THEN 810 ELSE IFN3$<>N1$ THEN GOTO 870
810 IF N4$="" THEN 820 ELSE IFN4$<>N2$ THEN GOTO 870
820 IF S1$="" THEN 830 ELSE IFS1$<>S$ THEN GOTO 870
830 IF C1$="" THEN 840 ELSE IFC1$<>C$ THEN GOTO 870
840 IF P3$="" THEN 850 ELSE IFP3$<>P1$ THEN GOTO 870
850 IF P4$="" THEN 860 ELSE IFP4$<>P2$ THEN GOTO 870
860 GOSUB 890:PRINT@456,USING"###";RECORD;:GOSUB520:RETURN
870 GOTO780
880 'PRINT RECORD ON SCREEN
890 PRINT@11,N1$;
900 PRINT@43,N2$;
910 PRINT@75,S$;
920 PRINT@107,C$;
930 PRINT@138,P1$;
940 PRINT@149,P2$;
950 RETURN
960 'PREPARE FOR NEW RECORD
970 A$ = STRING$(20,"-"):N1$ = A$:N2$ = A$:S$ = A$:C$ = A$:P1$ =
"-----":P2$ = "-----"
980 RECORD=R1
990 RETURN
1000 'FIND FILE SIZE OR CREATE IF NONE
1010 OPEN"D",#1,"DATABASE/DAT:0",100
1020 LF=LOF(1)
1030 CLOSE#1
1040 RETURN
1050 'OPEN FILE TO 'PUT' DATA INTO IT
1060 OPEN"D",#1,"DATABASE/DAT:0",100
1070 FIELD #1,20 AS SN$,20 AS CN$,20 AS ST$,20 AS CT$,4 AS PC$,10
AS PH$,6 AS SP$
1080 LSET SN$=N1$
1090 LSET CN$=N2$

```

```

1100 LSET ST$=S$
1110 LSET CT$=C$
1120 LSET PC$=P1$
1130 RSET PH$=P2$
1140 LSET SP$=STRING$(6,"-")
1150 PUT#1,RECORD
1160 CLOSE#1
1170 RETURN
1180 'OPEN FILE TO 'GET' DATA FROM IT
1190 OPEN"D",#1,"DATABASE/DAT:0",100
1200 FIELD #1,20 AS SN$,20 AS CN$,20 AS ST$,20 AS CT$,4 AS PC$,10
AS PH$,6 AS SP$
1210 GET#1,RECORD
1220 N1$=SN$:N2$=CN$:S$=ST$:C$=CT$:P1$=PC$:P2$=PH$
1230 CLOSE#1
1240 RETURN

```

Pill By Rick

```

0 'IF YOUR COCO CAN'T HANDLE THEHIGH SPEED POKE, DELETE LINE 101
BEFORE RUNNING.
1 GOTO100
2 FORY=0TO8:PMODE3:GET(4,9-Y)-(28,18-Y),P:PMODE3,2:PUT(120,0)-
(144,10),P:PMODE3:GET(10,1)-(18,8),P:GET(10,9)-(18,9),L
3 PUT(10,2)-(20,9),P:PUT(10,1)-(20,1),L
4 RT=TIMER:IFRT=TR THEN4ELSETR=RT
5 NEXT:FORY=0TO192
6 PMODE3:GET(4,0)-(28,9),P:PMODE3,2:PUT(120,Y)-
(144,Y+9),P:PMODE3:GET(10,1)-(20,8),P:GET(10,9)-(20,9),L
7 PUT(10,2)-(20,9),P:PUT(10,1)-(20,1),L:IFINKEY$>" THEN999
8 RT=TIMER:IFRT=TR THEN8ELSETR=RT
9 NEXT:GOTO2
10 DATA 0,1,1,1,1,1,1,1,1,1,0
20 DATA 1,1,1,0,4,4,6,0,1,1,1
30 DATA 1,1,1,0,4,0,4,0,1,1,1
40 DATA 1,1,1,0,4,0,4,0,1,1,1
50 DATA 1,1,1,0,4,4,6,0,1,1,1
60 DATA 1,1,1,0,4,5,6,0,1,1,1
70 DATA 1,1,1,0,4,0,4,0,1,1,1
80 DATA 1,1,1,0,4,0,4,0,1,1,1
90 DATA 0,1,1,1,1,1,1,1,1,1,0
100 PMODE0:PCLEAR5:DIM
P(20,9),L(18,1):P1(1)=2:P1(0)=1:P1(2)=2:P1(4)=0:P1(5)=2:P1(6)=3
101 IFPEEK(33021)=50THENPOKE65497,0ELSEPOKE65495,0
102 PMODE3,2:PCLS2:COLOR0:LINE(0,0)-(254,8),PSET,BF:PMODE4,2
103
DRAW"BM86,0;C0;R1D6L1R2U6R2D1R1D1L1D1;NL1;D1R1D1L1D1L1;BR5;U6R1D6
R4;BR3BU1;U5R1D6R3U6R1D5;BD1BR3"

```

```
104 DRAW"NU6;R1;NR4;U3;NR2;U3R4;BR3;BR10;BD6"
105
DRAW"BR3;L1U1L1U1L1U2R1;ND1;U1R1U1R1;BR3;ND2;R1D3R1D3R1U3R1U3R1D2
;BR3BD4"
106
DRAW"R1U2R1D1U2R1U2R1D1U2R1;BR3;ND6;R1D2;ND4;R1D1R1D1R1;NU4;D2R1U
6;BR3;R1D1R1D1R1D2L1;NU1;D1L1D1L1;BR6;BR2;R1;BU2;L1U1R2U1R1U1L1U1
L3D1L1"
107 SCREEN1,1
108 A$=INKEY$:IFA$="N"THEN111
109 IFA$="Y"THENP1(1)=3:GOTO111
110 GOTO108
111 PMODE3,2:PCLS1:PMODE0:PCLS1:PMODE3:PCLS
112 FOR Y=0 TO 8:FOR X=0 TO 20 STEP
2:READS:PSET(X+4,Y,P1(S)):NEXT X,Y
113 GET(4,0)-(26,8),P,G:LINE(6,0)-(24,0),PRESET:PUT(4,1)-
(26,9),P,PSET
114 PMODE4,2:PCLS:TR=TIMER:GOTO2
999 POKE65496,0:POKE65494,0:CLEAR:END:RUN
1997 'A SIMPLE PROGRAM I DID WAYBACK IN THE EARLY 1990S.
BEFORERELEASING IT, I TRANSFERRED ALL TEXT TO THE GRAPHICS
SCREENS, AND MAKE IT REQUIRE A PCLEAR 5 INSTEAD OF A PCLEAR 8.
1998 'ENTER "DEL 101" BEFORE YOURUN THIS PROGRAM IF YOUR COCO
CAN'T HANDLE THE HIGH SPEED POKE(POKE 65495,0).
1999 'PUBLIC DOMAIN 11/30/2004 - RICHARD KELLY. E-MAIL ME AT:
ARBORPA@YAHOO.COM
```